BATXILERGOKO MATEMATIKA MAPLEREKIN II

Fernando Garatea

Este artículo corresponde a la segunda parte del artículo titulado "Batxilergoko Matematika Maplerekin", publicado en la revista SIGMA, número 23.

10. INTEGRALAK

10.1. Ariketa

Idatzi ondoko baldintzak betetzen dituen *f (x)* funtzioa:

$$\begin{cases} f''(x) = e^{\frac{x}{2}} \sin(2x) \\ f(0) = 1 \\ f'(\pi) = 2 \end{cases}$$

10.1.1. Ebazpena

Idazketa argitzeko, funtzioa *f-z* adieraziko dugu, lehenengo deribatua *f1*-ez eta bigarrena *f2*-z.

Eta jakin nahi dugu bigarren deribatutzat $f_2(x) = e^{\frac{x}{2}} \sin(2x)$ adierazpena duten *f (x)* guztietatik zeinek betetzen *dituen ondoko baldintzak: f (0) = 1* eta *f1 (Pi) = 2*. Has gaitezen *f, f1* eta *f2* definitzen:

> **f2: = x - > exp (x / 2)*sin (2*x);**

$$f2 := x \to \mathbf{e}^{(1/2\,x)} \sin(2\,x)$$

Bigarren deribatua definitu dugu. Lehenengo deribatua, bigarren deribatuaren jatorrizkoa modura definituko dugu:

> **int (f2 (x), x);**

$$-\frac{8}{17}\mathbf{e}^{(1/2\,x)}\cos(2\,x) + \frac{2}{17}\mathbf{e}^{(1/2\,x)}\sin(2\,x)$$

Aurreko lerroan jatorrizko bakarra lortu dugu, baina jatorrizko guztiak *x* eta *k1* aldagaien menpe dagoen ondoko funtzioak emango dizkigu :

> **f1: = (x, k1) - > - 8/17*exp (1/2*x) *cos (2*x) + 2/17*exp (1/2*x) *sin (2*x) + k1;**

$$f1 := (x, k1) \to -\frac{8}{17}\mathbf{e}^{(1/2\,x)}\cos(2\,x) + \frac{2}{17}\mathbf{e}^{(1/2\,x)}\sin(2\,x) + k1$$

Eta *f* lehenengo deribatuaren jatorrizkotzat definituko dugu:

> **int (f1 (x, k1), x);**

$$-\frac{32}{289}\mathbf{e}^{(1/2\,x)}\cos(2\,x) - \frac{60}{289}\mathbf{e}^{(1/2\,x)}\sin(2\,x) + k1\,x$$

Lehenengo deribatuaren jatorrizko guztiak ondoko funtzioak ematen ditu:

> **f: = (x, k1, k2) - > - 32/289*exp (1/2*x) *cos (2*x) - 60/289*exp (1/2*x) *sin (2*x) + k1*x + k2;**

$$f := (x, k1, k2) \to -\frac{32}{289}\mathbf{e}^{(1/2\,x)}\cos(2\,x) - \frac{60}{289}\mathbf{e}^{(1/2\,x)}\sin(2\,x) + k1\,x + k2$$

Ondoren, funtzioari *f(0) = 1* eta $f1(\pi) = 2$ baldintzak ezarriko dizkiogu:

> **f (0, k1, k2) = 1;**

$$-\frac{32}{289} + k2 = 1$$

> **f1 (Pi, k1) = 2;**

$$-\frac{8}{17}\mathbf{e}^{(1/2\,\pi)} + k1 = 2$$

Non betetzen dira baldintza biak batera? Sistemaren soluzioetan:

> **solve ({-32/289 + k2 = 1,-8/17*exp (1/2*Pi) + k1 = 2}, {k1,k2});**

$$\left\{k2 = \frac{321}{289},\, k1 = \frac{8}{17}\mathbf{e}^{(1/2\,\pi)} + 2\right\}$$

Beraz, baldintza biak betetzen dituen funtzioa ondoko hauxe da:

> **f(x): = f (x,8/17*exp (1/2*Pi) + 2,321/289);**

$$\mathrm{f}(x) := -\frac{32}{289}\mathbf{e}^{(1/2\,x)}\cos(2\,x) - \frac{60}{289}\mathbf{e}^{(1/2\,x)}\sin(2\,x) + \left(\frac{8}{17}\mathbf{e}^{(1/2\,\pi)} + 2\right)x + \frac{321}{289}$$

10.2. Ariketa

$\int_{-4}^{4}(x + 2)(x - 1)(x - 3)$ integral mugaturantz hurbiltzeko, [-4,4] tartea zabalera bereko *10* azpitartetan bananduko dugu; ondoren, azpitarte horien erdiguneetan integrakizunaren balioak hartzen dituen funtzio mailakatua eraikiko dugu, eta gero, funtzio mailakatuaren eta *OX* ardatzaren artean eratutako laukizuzen azaleren baturaren bidez hurbilduko gatzaizkio integral mugatuari. Egizu. Izan daiteke negatiboa integral mugatua?

10.2.1 Ebazpena

Defini dezagun funtzioa:

> **f: = x - > (x + 2) * (x - 1) * (x - 3);**

$$f := x \rightarrow (x+2)(x-1)(x-3)$$

eta irudikatu:

> **plot (f (x), x = -4..4);**



Funtzio horren integral mugaturantz hurbiltzeko [-4,4] tartean, tartea zabalera bereko 10 azpitartetan banandu eta funtzio mailakatu bat eraikiko dugu, halako eran non azpitarteen erdiguneetan funtzio mailakatuak *f(x)*-ren balioak hartzen dituen. Hori egiteko, *student* paketeko *middlebox* komandoa daukagu:

> **with(student):**

> **middlebox (f (x), x = -4..4,10);**



Eta orain eratutako laukizuzenen azaleren batura kalkulatuko dugu, baina jakinda *OX* ardatzaren gainetik dauden laukizuzenen azaleren baturari ardatzaren azpitik daudenena kenduko diola:

> **middlesum (f (x), x = -4..4,10);**

$$\frac{4}{5}\left(\sum_{i=0}^{9}\left(-\frac{8}{5}+\frac{4}{5}i\right)\left(-\frac{23}{5}+\frac{4}{5}i\right)\left(-\frac{33}{5}+\frac{4}{5}i\right)\right)$$

> **evalf (%);**

-36.48000000

Emaitza horrek integral mugatua negatiboa dela iradokitzen du. Integral mugatua izan daiteke negatiboa? Noski, funtzioak abzisen ardatzaren azpitik eratzen duen eremuaren azalera gainetik eratzen duenarena baino handiagoa bada, negatiboa da integral mugatua. *OX*-en gainetik eratutakoa azpitik eratutakoa baino handiagoa bada, positiboa. Eta bi eremuak azalera berekoak badira, zero da integral mugatua.

Gure kasukoa negatiboa dela egiaztatzeko, kalkula dezagun integral mugatua:

> **int (f (x), x = -4..4);**

$$\frac{-112}{3}$$

> **evalf (%);**

-37.33333333

egiaztatu nahi genuenez.

11. MATRIZEAK

11.1. ARIKETA

Ebatz ezazu $\begin{cases} 2A+3B &= P \\ 3A-4B &= Q \end{cases}$ sistema matriziala, non P eta Q ondoko matrizeak diren:

$$P = \begin{bmatrix} 2 & 7 \\ -8 & 13 \\ 15 & 6 \end{bmatrix}, \quad Q = \begin{bmatrix} 3 & 2 \\ 5 & -6 \\ -20 & 9 \end{bmatrix}$$

Ondoren,

a) Kalkulatu $C = AB^T + I$ eta $E = BA^T + I$ matrizeak, non I identitate matrizea den, B^T, B-ren iraulia eta A^T, A-ren iraulia.

b) Kalkulatu $F = 2C + 3E$ matrizea.

c) Isolatu X matrizea $CXE - 3F = G$ ekuaziotik, G = $\begin{bmatrix} 71 & -39 & 39 \\ -45 & 106 & 102 \\ 33 & 144 & 159 \end{bmatrix}$ matrizea izanik.

d) Kalkulatu, azkenez, X^{108} berredura.

11.1.1. Ebazpena

P eta *Q* matrizeak definitzen hasiko gara:

> **P: = matrix (3,2,[2,7,-8,13,15,6]);**

$$P := \begin{bmatrix} 2 & 7 \\ -8 & 13 \\ 15 & 6 \end{bmatrix}$$

> **Q: = matrix (3,2,[3,2,5,-6,-20,9]);**

$$Q := \begin{bmatrix} 3 & 2 \\ 5 & -6 \\ -20 & 9 \end{bmatrix}$$

Eta, ondoren, sistema ebatziko dugu:

> **solve ({2*A + 3*B = P, 3*A - 4*B = Q}, {A,B});**

$$\{A = \frac{4}{17}P + \frac{3}{17}Q, B = \frac{3}{17}P - \frac{2}{17}Q\}$$

Horra hor *A* eta *B* matrizeak *P* eta *Q* matrizeen funtzioan. Taula formatoan idatziko ditugu:

> **A: = evalm (4/17*P + 3/17*Q);**

$$A := \begin{bmatrix} 1 & 2 \\ -1 & 2 \\ 0 & 3 \end{bmatrix}$$

> **B: = evalm (3/17*P - 2/17*Q);**

$$B := \begin{bmatrix} 0 & 1 \\ -2 & 3 \\ 5 & 0 \end{bmatrix}$$

a) atala

Behin sistema ebatzita, ariketan eskatzen dizkiguten gainontzeko elementuak kalkulatuko ditugu. $C = AB^T + I$ eta $E = BA^T + I$ matrizeak kalkulatzeko, lehendabizi I definitu behar dugu, eta AB^T zein BA^T matrizeak (3,3) ordenakoak direnez, horiekin batu daitekeen identitate matrizeak ere (3,3) ordenakoa izan behar du. I3 deituko diogu.

> w**ith (linalg):**

> **I3: = diag (1,1,1);**

$$I3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

> **C: = evalm (A&*transpose(B) + I3);**

$$C := \begin{bmatrix} 3 & 4 & 5 \\ 2 & 9 & -5 \\ 3 & 9 & 1 \end{bmatrix}$$

> **E: = evalm (B&*transpose(A) + I3);**

$$E := \begin{bmatrix} 3 & 2 & 3 \\ 4 & 9 & 9 \\ 5 & -5 & 1 \end{bmatrix}$$

b) atala

Kalkula dezagun, orain, $F=2C + 3E$ matrizea:

> **F: = evalm (2*C + 3*E);**

$$F := \begin{bmatrix} 15 & 14 & 19 \\ 16 & 45 & 17 \\ 21 & 3 & 5 \end{bmatrix}$$

c) atala

$CXE - 3F = G$ ekuaziotik X isolatzeko, $CXE = G + 3F$ egingo dugu lehenengo eta behin. Ondoren, C-ren alderantzizkoak bidertuko ditugu ekuazioaren bi aldeak, ezkerretik: $C^{-1}CXE = C^{-1}(G + 3F)$ eta $C^{-1}C = I$ denez, $XE = C^{-1}(G + 3F)$ izango dugu. Orain, E-ren alderantzizkoaz bidertuko ditugu ekuazioaren bi aldeak, eskuinetik: $XEE^{-1} = C^{-1}(G + 3F)\,E^{-1}$ lortuz, eta hortik $E\,E^{-1} = I$ dela kontuan izanda X bakanduko dugu: $X = C^{-1}(G + 3F)\,E^{-1}$. Baina badaukagu berdintza horretan definitu gabeko matrize bat: G. Defini dezagun:

> **G: = matrix (3,3,[71,-39,39,-45,106,102,33,144,159]);**

$$G := \begin{bmatrix} 71 & -39 & 39 \\ -45 & 106 & 102 \\ 33 & 144 & 159 \end{bmatrix}$$

> **X: = evalm (inverse (C) &* (G + 3*F) &*inverse (E));**

$$X := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

d) atala

X^{108} kalkulatzea besterik ez zaigu geratzen. Zuzenean eska dezakegun arren X^{108} kalkulatzeko, X^n egiten ahaleginduko gara lehenengo, eta gero, *n=108* ordeztuko dugu X^n-ren formulan, X^{108} lortzeko. Idatz ditzagun, bada, *X*-en lehenengo berredurak, X^n-rentzako lege orokorra topatzeko:

> **evalm (X), evalm (X^2), evalm (X^3), evalm (X^4);**

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 3 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 4 & 0 \\ 0 & 0 & 9 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 8 & 0 \\ 0 & 0 & 27 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 16 & 0 \\ 0 & 0 & 81 \end{bmatrix}$$

Dirudienez, X^n diagonaleko elementuak *ber* n eginda lortzen da, hots,

$$\begin{bmatrix} 1^n & 0 & 0 \\ 0 & 2^n & 0 \\ 0 & 0 & 3^n \end{bmatrix}$$

Orduan, X^{108} diagonaleko elementuak ber 108 eginda aterako dugu:

$$X^{108} = \begin{bmatrix} 1^{108} & 0 & 0 \\ 0 & 2^{108} & 0 \\ 0 & 0 & 3^{108} \end{bmatrix}$$

Mapleri eragiketa egiteko eskatuz gero, hara hemen emaitza:

> **evalm (X^108);**

$$\begin{bmatrix} 1, & 0, & 0 \\ 0, & 324518553658426726783156020576256, & 0 \\ 0, & 0, & 3381391913522726342930221472392241170198527451848561 \end{bmatrix}$$

11.2. ARIKETA

Idatzi 3 ordenako matrize karratu bat, determinanteak 6 balio duela eta elementuen artean 0-rik ez dagoela jakinik. Ondoren, *swapcol* edo *swaprow* komandoen laguntzaz trukatu bi errenkada paralelo edo bi zutabe paralelo eta kalkulatu matrize berriaren determinantea. Zein da ondorioa?

11.2.1. Ebazpena

Hiru ordenako matrize karratu bat eraiki behar dugu, determinantea 6 eta elementu guztiak ez-nuluak dituela jakinik. Horretarako, diagonaleko elementuen biderkadura 6 duen matrize goi-triangeluar batetik abiatuko gara, gero, errenkadei eta zutabeei gainontzekoen konbinazio linealak batuta, matrizea zerorik gabe uzteko.

> **with (linalg):**

> **A: = matrix ([[1,2,3], [0,2,4], [0,0,3]]);**

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 0 & 2 & 4 \\ 0 & 0 & 3 \end{bmatrix}$$

> **det (A);**

$$6$$

> **addrow (A,1,2,2);**

$$\begin{bmatrix} 1 & 2 & 3 \\ 2 & 6 & 10 \\ 0 & 0 & 3 \end{bmatrix}$$

> **addrow (%,1,3,-2);**

$$\begin{bmatrix} 1 & 2 & 3 \\ 2 & 6 & 10 \\ -2 & -4 & -3 \end{bmatrix}$$

> **det (%);**

$$6$$

> **B: = %%;**

$$B := \begin{bmatrix} 1 & 2 & 3 \\ 2 & 6 & 10 \\ -2 & -4 & -3 \end{bmatrix}$$

B matrizeak betetzen ditu baldintza guztiak. Orain, *B* horretan, errenkadak eta zutabeak elkarren artean trukatzeari ekingo diogu, determinantea zelan aldatzen den ikusteko:

> **swaprow (B,1,2);**

$$\begin{bmatrix} 2 & 6 & 10 \\ 1 & 2 & 3 \\ -2 & -4 & -3 \end{bmatrix}$$

> **det (%);**

-6

> **swapcol (B,1,3);**

$$\begin{bmatrix} 3 & 2 & 1 \\ 10 & 6 & 2 \\ -3 & -4 & -2 \end{bmatrix}$$

> **det (%);**

-6

Zein da ondorioa? Errenkada bi elkarren artean trukatuz gero, determinantearen balioa zeinuz aldatzen dela. Eta zutabe bi elkarren artean trukatuz gero, beste horrenbeste.

12. SISTEMAK

12.1. ARIKETA

Aztertu $\begin{cases} x + ay - z = 1 \\ 2x + y - az = 2 \\ x - y - z = a - 1 \end{cases}$ planoen posizioa, a parametroaren balioen arabera.

12.1.1. Ebazpena

Hiru planoen arteko posizioa aztertu aurretik, sistemaren bateragarritasuna aztertuko dugu, *a* parametroaren balioen arabera. Defini ditzagun, beraz, ekuazioak, *linalg* paketea indarrean jarrita:

> **with (linalg):**

Warning, the name inverse has been redefined

Warning, the protected names norm and trace have been redefined and unprotected

> **ek1: = x + a*y - z = 1:**

> **ek2: = 2*x + y - a*z = 2:**

> **ek3: = x - y - z = a -1:**

Ondoren, sistemaren matrize zabaldua pantailaratuko dugu, *genmatrix* komandoarekin:

> **A: = genmatrix ([ek1,ek2,ek3], [x,y,z], flag);**

$$A := \begin{bmatrix} 1 & a & -1 & 1 \\ 2 & 1 & -a & 2 \\ 1 & -1 & -1 & a-1 \end{bmatrix}$$

Eta mailakatu egingo dugu, **A** matrizeari *ffgausselim* komandoa aplikatuz:

> **ffgausselim (A);**

$$\begin{bmatrix} 1 & a & -1 & 1 \\ 0 & 1-2a & -a+2 & 0 \\ 0 & 0 & a-a^2+2 & 5a-2a^2-2 \end{bmatrix}$$

Lerroburuak zero eginez, heina alda dezaketen parametroaren balioak erdietsiko ditugu:

> **solve ((1 - 2*a) * (a - a^2 + 2) = 0,a);**

$$\frac{1}{2}, 2, -1$$

Orduan, *a* parametroa *1/2, 2,* edo *-1* ez denean, sistema bateragarri determinatua da, soluzio bakarra duena. Eta horrek geometrikoki zer diosku? Bada, planoak tetraedro baten aurpegien gain etzanda daudela, puntu bakarrean elkar ebakiz

> **with (geom3d):**

> **tetrahedron (t, point (o,1,2,3), 3);**

t

> **draw (t);**



Orain, *a = 1/2* hartuko dugu eta *A* matrizean ordezkatu:

> **subs (a = 1/2, evalm (A));**

$$\begin{bmatrix} 1 & \frac{1}{2} & -1 & 1 \\ 2 & 1 & \frac{-1}{2} & 2 \\ 1 & -1 & -1 & \frac{-1}{2} \end{bmatrix}$$

Mailakatuz:

> **ffgausselim (%);**

$$\begin{bmatrix} 1 & \frac{1}{2} & -1 & 1 \\ 0 & \frac{-3}{2} & 0 & \frac{-3}{2} \\ 0 & 0 & \frac{-9}{4} & 0 \end{bmatrix}$$

Beraz, kasu honetan ere, sistema bateragarri determinatua da, ekuazio aske kopurua (3), ezezagun kopuruarekin bat baitator. Orduan, hiru planoek puntu batean ebakiko dute elkar.

Har dezagun, ondoren, $a = 2$ eta A matrizean ordezkatu:

> **subs (a = 2, evalm (A));**

$$\begin{bmatrix} 1 & 2 & -1 & 1 \\ 2 & 1 & -2 & 2 \\ 1 & -1 & -1 & 1 \end{bmatrix}$$

Mailakatuz:

> **ffgausselim (%);**

$$\begin{bmatrix} 1 & 2 & -1 & 1 \\ 0 & -3 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Kasu honetan, sistema bateragarri inderteminatua da, lerro aske kopurua (*2*) ezezagun kopurua (*3*) baino txikiagoa baita. Geometrikoki, beraz, hiru planoek zuzen batean elkar ebakitzen dute. Binakako posizioa aztertuz, *p1* eta *p2* planoek zuzen batean ebakitzen dute elkar, *p1* eta *p3* planoek ere bai, eta *p2* eta *p3* planoek beste horrenbeste. Beraz, liburu bateko orriak balira bezala har ditzakegu. Grafikoki ikusteko, defini ditzagun hiru planoak, eta *draw* komandoarekin irudikatu:

> **plane (p1, x + 2*y - z = 1, [x,y,z]):**

> **plane (p2, 2*x + y - 2*z = 2, [x,y,z]):**

> **plane (p3, x - y - z = 2, [x,y,z]):**

> **draw ([p1, p2, p3]);**



Azkenez, $a = -1$ ordeztuko dugu A matrizean, emaitzeko matrizea mailakatuz:

> **subs (a = -1, evalm (A));**

$$\begin{bmatrix} 1 & -1 & -1 & 1 \\ 2 & 1 & 1 & 2 \\ 1 & -1 & -1 & -2 \end{bmatrix}$$

> **ffgausselim (%);**

$$\begin{bmatrix} 1 & -1 & -1 & 1 \\ 0 & 3 & 3 & 0 \\ 0 & 0 & 0 & -9 \end{bmatrix}$$

Eta sistema bateraezina da, kasu honetan, azken errenkadak adierazten duen *0=-9* ekuazioa beteko duen *(x,y,z)* hirukoterik ez dagoelako. Planoek, beraz, ez dute puntu komunik. Euren posizioa binaka aztertuz: *p1* eta *p3* paraleloak dira eta *p2-k* zuzen banatan ebakitzen ditu, baina hirurena den punturik ez dago. Grafikoki adierazteko, jarraitu ondoko urratsak:

> **with (geom3d):**

Warning, the name inverse has been redefined.

> **plane (p1, x - y - z = 1, [x,y,z]):**

> **plane (p2, 2*x + y + z = 2, [x,y,z]):**

> **plane (p3, x - y - z = -2, [x,y,z]):**

> **draw ([p1, p2, p3]);**

12.2. Ariketa

Definitu prozedura bat, zeinaren bidez Maplek parametro bakarreko sistemen azterketa eta ebazpena egiten dituen, eta pantailaratu emaitzak. Ondoren, probatu prozedura ondoko sistemekin:

$$a)\begin{cases} x+2y+3z = 0 \\ x+my+z = 0 \\ 2x+3y+4z = 2 \end{cases},\quad b)\begin{cases} x+my+z = 4 \\ x+3y+z = 5 \\ mx+y+z = 4 \end{cases},\quad c)\begin{cases} x+2z = 3 \\ 3x+y+z = -1 \\ 2y-z = -2 \\ x-y+mz = -5 \end{cases}$$

12.2.1. Ebazpena

Parametrorik gabeko sistemen azterketarako eraiki dugun *sisebaz* prozedurak hemen ere izango du egitekorik; beraz, prozedura hori berreskuratuz hasiko dugu ariketa:

```
> with (linalg):
> sisebaz: = proc (ek::list, ez::list):
> A: = genmatrix (ek, ez):
> B: = genmatrix (ek, ez, flag):
> if rank (A) = rank (B) and rank (B) = nops (ez) then print (`Sistema bateragarri determinatua da eta bere soluzio bakarra ondokoa:`), print (solve ({op (ek)}, {op (ez)})) elif rank (A) = rank (B) and rank (B)<nops (ez) then print (`Sistema bateragarri indeterminatua da eta bere infinitu soluzioak ondokoak:`), print (solve ({op (ek)}, {op (ez)})) else print (`Sistema bateraezina da, eta beraz, ez du soluziorik`) fi:
> end:
```

Eta, orain, parametrodun sistemen azterketarako balio duen prozedura idazteari ekingo diogu:

Warning, `A` is implicitly declared local to procedure `sisebaz`

Warning, `B` is implicitly declared local to procedure `sisebaz`

```
> azter: = proc (ek::list, ez::list):
> M: = genmatrix (ek, ez, flag):
> print (`Parametroaren balio erreal guztietarako`), sisebaz (ek, ez):
> LUdecomp (M, U1 = 'u1', det = 'd', rank = 'heina'):
> s: = solve (d):
> if nops ([s])> = 1 then print (`salbu parametroaren`, {s}, `balioetarako`) fi:
> if nops ([s]) = 1 then print (`Parametroa`, s, `denean`), sisebaz (subs (a = s, ek), ez) fi:
> if nops ([s])> = 2 then for i while i< = nops ([s]) do print (`Parametroa`, s[i], `denean`),
  sisebaz (subs (a = s[i], ek), ez) od: fi:
> end:
```

Warning, `M` is implicitly declared local to procedure `azter`

Warning, `s` is implicitly declared local to procedure `azter`

Warning, `i` is implicitly declared local to procedure `azter`

Egin dezagun egiaztapena pare bat sistemarekin:

```
> ek1: = x + 2*y + 3*z = 0:
> ek2: = x + a*y + z = 0:
> ek3: = 2*x + 3*y + 4*z = 2:
> azter ([ek1, ek2, ek3], [x, y, z]);
```

Parametroaren balio erreal guztietarako
Sistema bateragarri determinatua da eta bere soluzio bakarra ondokoa:

$$\{y = -2\,\frac{1}{-1+a},\; x = \frac{-2+3\,a}{-1+a},\; z = -\frac{-2+a}{-1+a}\}$$

salbu parametroaren, {1}, balioetarako
Parametroa, 1, denean
Sistema bateraezina da, eta beraz, ez du soluziorik

> **ek4: = x + a*y + z = 4:**

> **ek5: = x + 3*y + z = 5:**

> **ek6: = a*x + y + z = 4:**

> **azter ([ek4, ek5, ek6], [x, y, z]);**

Parametroaren balio erreal guztietarako
Sistema bateragarri determinatua da eta bere soluzio bakarra ondokoa:

$$\{z = \frac{-11 + 5\,a}{a - 3}, x = -\frac{1}{a - 3}, y = -\frac{1}{a - 3}\}$$

salbu parametroaren, {1, 3}, balioetarako
Parametroa, 1, denean
Sistema bateragarri indeterminatua da eta bere infinitu soluzioak ondokoak:

$$\{x = x, z = -x + \frac{7}{2}, y = \frac{1}{2}\}$$

Parametroa, 3, denean
Sistema bateraezina da, eta beraz, ez du soluziorik

> **ek7: = x + 2*z = 3:**

> **ek8: = 3*x + y + z = -1:**

> **ek9: = 2*y - z = -2:**

> **ek10: =x - y + a*z = -5:**

> **azter ([ek7, ek8, ek9, ek10], [x, y, z]);**

Parametroaren balio erreal guztietarako
Sistema bateraezina da, eta beraz, ez du soluziorik
salbu parametroaren, {-2}, balioetarako
Parametroa, -2, denean
Sistema bateragarri,determinatua da eta bere soluzio bakarra ondokoa:

$\{y = 0, z = 2, x = -1\}$

13. GEOMETRIA PLANOAN

13.1. ARIKETA

A(-1,2) eta *C(7,4)* puntuak *ABCD* karratuaren aurkako erpinak dira. Ediren beste bi erpinen koordenatuak. Frogatu aldeak binaka paraleloak eta binaka perpendikularrak direla. Eta, azkenez, irudikatu karratua eta karratuaren zirkunferentzia inskribatua nahiz zirkunskribatua.

13.1.1. Ebazpena

Hasteko, puntuak definituko ditugu, *geometry* paketea ireki ondoren:

> **with (geometry):**

> **point (A, -1,2), point (C, 7, 4);**

A, C

Aurkako erpinak ezagututa, MakeSquare komandoarekin definituko dugu karratua, ondoko eran:

> **MakeSquare (K, [A, C, 'diagonal']);**

K

K karratuaren ezaugarrietan *(detail* komandoak ematen ditu), beste bi erpinen koordenatuak agertuko dira:

> **detail (K);**

name of the object: K
form of the object: square2d
the four vertices of the square: [[-1, 2], [4, -1], [7, 4], [2, 7]]
the length of the diagonal: 68 **^***(1/2)*

Ikusten dugunez, beste bi erpinak *B(4,-1)* eta *D(2,7)* dira. Aldeen paralelotasuna eta perpendikulartasuna aztertzeko, defini ditzagun falta izan diren erpinak eta aldeak:

> **point (B, 4, -1), point (D, 2, 7);**

B, D

> **line (AB, [A,B]), line (BC, [B,C]), line (CD, [C,D]), line (DA, [D,A]);**

AB, BC, CD, DA

Orduan, *AB* eta *CD* elkarrekiko paraleloak direla frogatzeko, *AreParallel* komandoa erabiliko dugu, eta, *BC* eta *DA* paraleloak direla frogatzeko ere bai:

> **AreParallel (AB, CD);**

true

> **AreParallel (BC, DA);**

true

Ondoren, ondoko bikoteak elkarrekiko perpendikularrek direla frogatuko dugu *ArePerpendicular* komandoarekin: *AB* eta *BC, BC* eta *CD, CD* eta *DA, DA* eta *AB*.

> **ArePerpendicular (AB, BC);**

true

> **ArePerpendicular (BC, CD);**

true

> **ArePerpendicular (CD, DA);**

true

> **ArePerpendicular (DA, AB);**

true

Azkenez, zirkunferentzia inskribatuari eta zirkunskribatuari helduko diegu. Bataren zein bestearen zentroa *AC* segmentuaren erdigunean dago. Defini dezagun, orduan, *AC* segmentua segment komandoarekin:

> **segment (AC, [A,C]);**

$$AC$$

Eta, segmentua definituta, erdiko puntua *midpoint* komandoarekin lortuko dugu:

> **midpoint (E, AC);**

$$E$$

Zirkunferentzien zentroaren koordenatuak jakiteko, *coordinates* komandoa erabiliko dugu:

> **coordinates (E);**

$$[3, 3]$$

Ondoren, erradioen bila goaz. Zirkunferentzia inskribatuaren erradioa, begien bistakoa denez, karratuaren aldeak duen neurriaren erdikoa da; eta, zirkunskribatuarena, berriz, diagonalaren neurriaren erdikoa:

> **inskrierradioa: = distance (A, B) / 2;**

$$inskrierradioa := \frac{1}{2}\sqrt{34}$$

> **zirkunskrierradioa: = distance (A, C) / 2;**

$$zirkunskrierradioa := \frac{1}{2}\sqrt{68}$$

Zentroa eta erradioak ezagututa, defini ditzakegu zirkunferentziak:

> **circle (zirkunskribatua, [E, zirkunskrierradioa]);**

$$zirkunskribatua$$

> **circle (inskribatua, [E, inskrierradioa]);**

$$inskribatua$$

Eta, azkenez, karratua eta bi zirkunferentziak grafikoki adierazteko *draw* komandoa erabiliko dugu:

> **draw ([K, zirkunskribatua, inskribatua], view = [-2..8,-2..8]);**



14. GEOMETRIA ESPAZIOAN

14.1. Ariketa

A(0,1,0) eta *B(-1,1,1)* puntuak triangelu baten erpinetako bi dira; hirugarren erpina $r: \begin{cases} x = 4 \\ z = 1 \end{cases}$ zuzeneko *C* puntua da. Aurkitu *C* puntuaren koordenatuak, *A* eta *C* lotzen dituen zuzena *r* zuzenarekiko elkarzuta dela jakinik. Ondoren, kalkulatu triangeluaren ortozentroaren koordenatuak.

14.1.1. Ebazpena

Ariketan eman dizkiguten *A* zein *B* puntuak eta *r* zuzena definitzen hasiko gara:

> **with (geom3d):**

> **point (A, 0, 1, 0):**

> **point (B, -1, 1, 1):**

> **plane (p1, x = 4, [x,y,z]):**

> **plane (p2, z = 1, [x,y,z]):**

> **line (r, [p1, p2]):**

Pantailaratu ditzagun *r* zuzenaren ekuazio parametrikoak, zuzen horretakoa den *C* puntua definitu ahal izateko:

> **Equation (r,t);**

$$[4, -t, 1]$$

Beraz, *C (4, -t, 1)* da, eta r-ren norabide bektorea, berriz, *(0, -1,0)*.

Baina, *A* eta *C* lotzen dituen zuzena r-rekiko elkarzuta dela dio ariketak. Hortaz, defini dezagun *A* eta *C* lotzen dituen *AC* zuzena:

> **point (C, 4, -t, 1):**

> **line (AC, [A, C]);**

$$AC$$

> **Equation (AC, lambda);**

$$[4\,\lambda,\ 1 + (-t\ -1)\,\lambda,\ \lambda\,]$$

Orduan, *AC* zuzenaren norabide bektorea *(4, -t-1, 1)* da, eta *r*-ren norabide bektorearekiko elkarzuta da. Beraz, bi norabide bektore horien arteko biderkadura eskalarra zero izango da:

> **with (linalg):**

Warning, the name inverse has been redefined

> **AC: = vector ([4, -t-1, 1]):**

> **r_ren_norabidea: = vector ([0, -1, 0]):**

> **dotprod (AC, r_ren_norabidea) = 0;**

$$t + 1 = 0$$

eta biderkadura eskalarra zero izan dadin, *t=-1* behar dugu. Parametroaren balio hori *C* puntuan ordeztuz, *C*-ren koordenatuak izango ditugu: *(4,1,1)*. Defini dezagun *C*:

> **point (C, 4, 1, 1):**

Ondoren, ortozentroaren koordenatuak kalkulatu behar ditugu. Izan bitez *A1, A*-ren proiekzio ortogonala *B* eta *C* lotzen dituen zuzenaren gainean; *B1* puntua *B*-ren proiekzio ortogonala *A* eta *C* lotzen dituen zuzenean; *C1* puntua *C*-ren proiekzio ortogonala *A* eta *B* lotzen dituen zuzenaren gainean. Kalkula ditzagun **A1, B1** eta ***C1*** *projection* komandoarekin:

> **projection (A1, A, line (BC, [B,C])):**

> **projection (B1, B, line (AC, [A,C])):**

> **projection (C1, C, line (AB, [A,B])):**

Ortozentroa, *AA1, BB1* eta *CC1* altueren ebakidura da. Kalkulatzeko, idatz ditzagun altueron ekuazioak:

> **line (AA1, [A,A1]):**

> **line (BB1, [B,B1]):**

> **line (CC1, [C,C1]):**

Eta ikus dezagun hiru altuerek elkar ebakitzen dutela, *AreConcurrent* komandoarekin:

> **AreConcurrent (AA1, BB1, CC1);**

true

Ebakigunea kalkulatzeko, *intersection* komandoa aplikatuko diegu *AA1* eta *BB1* altuerei:

> **intersection (E, AA1, BB1);**

E

Eta *E* puntua *CC1* altuerakoa ere badela egiaztatuko dugu:

> **IsOnObject (E, CC1);**

true

Ortozentroaren koordenatuak, azkenez, ondokoak dira:

> **coordinates (E);**

[0, 1, -3]

15. KONBINATORIA

15.1. ARIKETA

Hiru dado bereiztezin mahai gainera jaurtiz gero, zenbat emaitza desberdin lortuko genituzke? Zeintzuk? Ba al dago hori automatizatzerik?

15.1.1. Ebazpena

Hiru dado bereiztezin mahai gainera jaurtiz gero, zenbat emaitza desberdin jasoko genituzke?

Dadoak bereiztezinak badira, *112, 121, 211* emaitzak bat dira, *345, 354, 435, 453, 534, 543* emaitzak ere bat. Orduan zenbat dira?

Zenbat diren esatea erraza da seiren errepikatuzko konbinazioak hirunaka hartuta:

> **with (combinat):**

> **binomial (6 + 3-1, 3);**

56

Zeintzuk dira 56 horiek?

Lehenengo, zerrendako elementuak idatziko ditugu, banaka harturiko errepikatuzko konbinazioak lortuz. Ondoren, horietako bakoitzari bera eta bera baino handiagoak erantsiko dizkiogu, banan- banan, binaka hartutako errepikatuzko konbinazioak lortuz. Eta aurrekoei azken zifra eta azken zifra baino handiagoak erantsiz lortuko ditugu hirunaka harturiko errepikatuzko konbinazioak.

Banakakoak, ondokoak:

> **J1: = [1, 2, 3, 4, 5, 6];**

J1: = [1, 2, 3, 4, 5, 6]

Ondoren, horietako bakoitzari bera eta bera baino handiagoak diren zenbakiak erantsiko dizkiogu, banan-banan, binaka hartutako errepikatuzko konbinazioak lortuz eta, bide batez, bi dado bereiztezinen jaurtiketaren emaitzak aterata:

> **J2: = [11, 12, 13, 14, 15, 16, 22, 23, 24, 25, 26, 33, 34, 35, 36, 44, 45, 46, 55, 56, 66];**

J2: = [11, 12, 13, 14, 15, 16, 22, 23, 24, 25, 26, 33, 34, 35, 36, 44, 45, 46, 55, 56, 66];

Eta aurrekoei azken zifra eta zifra hori baino handiagoak diren zenbakiak lotuz lortuko ditugu hirunaka harturiko errepikatuzko konbinazioak.

> **J3: = [111, 112, 113, 114, 115, 116, 122, 123, 124, 125, 126, 133, 134, 135, 136, 144, 145, 146, 155, 156, 166, 222, 223, 224, 225, 226, 233, 234, 235, 236, 244, 245, 246, 255, 256, 266, 333, 334, 335, 336, 344, 345, 346, 355, 356, 366, 444, 445, 446, 455, 456, 466, 555, 556, 566, 666];**

J3: = [111, 112, 113, 114, 115, 116, 122, 123, 124, 125, 126, 133, 134, 135, 136, 144, 145, 146, 155, 156, 166, 222, 223, 224, 225, 226, 233, 234, 235, 236, 244, 245, 246, 255, 256, 266, 333, 334, 335, 336, 344, 345, 346, 355, 356, 366, 444, 445, 446, 455, 456, 466, 555, 556, 566, 666];

Zenbat idatzi ditugu?

> **nops (J3);**

56

Behar genituenak. Baina ez al dago errepikatuzko konbinazioak automatizatzerik? Idatz dezagun prozedura bat hori egiteko. Lehenengo eta behin, bi multzoren cartesiarra emango digun *Bidercar* prozedura idatziko dugu, eta horrexetan oinarrituta idatziko dugu errepikatuzko konbinazioak emango dizkiguna:

> **Bidercar: = proc (A::list, B::list)**

> **K: = []:**

> **for i to nops (A) do for j to nops (B) do K: = [op (K), [op (op (i,A)), op (j,B)]] od: od:**

> **end:**

Biderketa cartesiarraren funtzionamendua ikusteko, biderka ditzagun *J1* eta *J1* zerrendak:

> **Bidercar (J1,J1);**

[[1, 1], [1, 2], [1, 3], [1, 4], [1, 5], [1, 6], [2, 1], [2, 2], [2, 3], [2, 4], [2, 5], [2, 6],
[3, 1], [3, 2], [3, 3], [3, 4], [3, 5], [3, 6], [4, 1], [4, 2], [4, 3], [4, 4], [4, 5], [4, 6],
[5, 1], [5, 2], [5, 3], [5, 4], [5, 5], [5, 6], [6, 1], [6, 2], [6, 3], [6, 4], [6, 5], [6, 6]
]

Idatz dezagun, ondoren, errepikatuzko konbinazioak emango dizkigun prozedura:

> **ErreKonbi: = proc (A::list, B::list)**

> **K: = []:**

> **for i to nops (A) do for j to nops (B) do if op (nops (op (i,A)), op (i,A)) = op (j,B) then K: = [op (K), op (Bidercar ([op (i,A)], [op (k,B) $k = j..nops (B)]))] fi: od: od:**

> **end:**

Eta binaka harturiko errepikatuzko konbinazioak idazteari ekingo diogu, ondoren:

> **ErreKonbi (J1, J1);**

[[1, 1], [1, 2], [1, 3], [1, 4], [1, 5], [1, 6], [2, 2], [2, 3], [2, 4], [2, 5], [2, 6], [3, 3],
[3, 4], [3, 5], [3, 6], [4, 4], [4, 5], [4, 6], [5, 5], [5, 6], [6, 6]]

Hirunaka harturiko errepikatuzko konbinazioak:

> **ErreKonbi (%, J1);**

[[1, 1, 1], [1, 1, 2], [1, 1, 3], [1, 1, 4], [1, 1, 5], [1, 1, 6], [1, 2, 2], [1, 2, 3], [1, 2, 4],
[1, 2, 5], [1, 2, 6], [1, 3, 3], [1, 3, 4], [1, 3, 5], [1, 3, 6], [1, 4, 4], [1, 4, 5], [1, 4, 6],
[1, 5, 5], [1, 5, 6], [1, 6, 6], [2, 2, 2], [2, 2, 3], [2, 2, 4], [2, 2, 5], [2, 2, 6], [2, 3, 3],
[2, 3, 4], [2, 3, 5], [2, 3, 6], [2, 4, 4], [2, 4, 5], [2, 4, 6], [2, 5, 5], [2, 5, 6], [2, 6, 6],
[3, 3, 3], [3, 3, 4], [3, 3, 5], [3, 3, 6], [3, 4, 4], [3, 4, 5], [3, 4, 6], [3, 5, 5], [3, 5, 6],

Eta horiexek dira hiru dado bereiztezin mahai gainera jaurtitzean lor daitezkeen emaitzak.

16. Estatistika

16.1. Ariketa

Euskal Autonomia Erkidegoko Bigarren Hezkuntzako Matematikako oposizioetan hiru epaimahia eratu ziren, lurralde bakoitzeko bana, eta emaitzak lurraldeka hauexek izan ziren: Araba=[3, 3, 3, 3, 3, 3, 3, 3, 4, 4, 4, 4, 4, 4, 5, 5, 5, 5, 5, 5, 6, 6, 6, 6, 6, 6, 6, 7, 7, 7, 7, 7, 7, 7, 8, 8, 8, 9, 9, 9, 10, 10], Bizkaia:=[1, 1.25, 1.5, 1.5, 1.5, 1.5, 1.5, 1.5, 1.5, 1.5, 1.5, 1.5, 1.75, 1.75, 1.75, 2, 2, 2, 2.75, 3, 3, 3, 3, 3, 3, 3.25, 3.25, 3.25, 3.25, 3.25, 3.25, 3.25, 3.5, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4.5, 4.5, 5, 5, 5, 5, 5, 5.5, 5.75, 6, 6, 6, 6, 6, 6, 6, 6, 6, 6, 6.25, 6.25, 6.5, 6.75, 6.75, 7, 7, 7.5, 8, 8, 8.25, 8.25, 8.5, 9, 10],

eta Gipuzkoa=[3, 3, 3, 3, 4, 4, 5, 5, 5, 5, 5, 5, 6, 6, 6, 6, 6, 6, 6, 6, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 8, 8, 8, 8, 8, 9, 9, 9, 9, 9, 9, 10, 10, 10, 10]

Azken sailkapena EAEn epaimahai bakarra egon izan balitz bezala egin zuten, eta horren arabera, nor sailkatuko zen arinago, Bizkaian 5 atera zuena, Araban 6 atera zuena ala Gipuzkoan 7 atera zuena?

16.1.1. Ebazpena

Datuak sartuko ditugu zerrendetan:

> **Araba:=[3, 3, 3, 3, 3, 3, 3, 3, 4, 4, 4, 4, 4, 4, 5, 5, 5, 5, 5, 5, 6, 6, 6, 6, 6, 6, 6, 7, 7, 7, 7, 7, 7, 7, 8, 8, 8, 9, 9, 9, 10, 10]:**

> **Bizkaia:=[1, 1.25, 1.5, 1.5, 1.5, 1.5, 1.5, 1.5, 1.5, 1.5, 1.5, 1.5, 1.75, 1.75, 1.75, 2, 2, 2, 2.75, 3, 3, 3, 3, 3, 3, 3.25, 3.25, 3.25, 3.25, 3.25, 3.25, 3.25, 3.5, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4.5, 4.5, 5, 5, 5, 5, 5, 5.5, 5.75, 6, 6, 6, 6, 6, 6, 6, 6, 6, 6, 6.25, 6.25, 6.5, 6.75, 6.75, 7, 7, 7.5, 8, 8, 8.25, 8.25, 8.5, 9, 10]:**

> **Gipuzkoa:=[3, 3, 3, 3, 4, 4, 5, 5, 5, 5, 5, 5, 6, 6, 6, 6, 6, 6, 6, 6, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 7, 8, 8, 8, 8, 8, 9, 9, 9, 9, 9, 9, 10, 10, 10, 10]:**

Ondoren, banaketa bakoitzaren batez bestekoa eta desbidazio estandarra kalkulatuko ditugu, *describe* azpipaketeko *mean* eta *standarddeviation* komandoekin:

> **with (stats [describe]);**

[*coefficientofvariation, count, countmissing, covariance, decile, geometricmean, harmonicmean, kurtosis, linearcorrelation, mean, meandeviation, median, mode, moment, percentile, quadraticmean, quantile, quartile, range, skewnwss, standarddeviation, sumdata, variance*]

> **mean (Araba), standarddeviation (Araba);**

$$\frac{40}{7}, \frac{1}{7}\sqrt{206}$$

> **evalf (%);** 5.714285714,2.050385728

> **mean (Bizkaia), standarddeviation (Bizkaia);**

4.347402597,2.193983753

> **mean (Gipuzkoa), standarddeviation (Gipuzkoa);**

$$\frac{337}{50}, \frac{1}{50}\sqrt{8781}$$ > **evalf (%);**

6.740000000, 1.874139803

Sailkapen orokorrean arinagokoa zein den kalkulatzeko, notari batez bestekoa kendu eta kenketa desbidazio estandarraz zatituko dugu. Zatiketaren emaitzak EAEko sailkapena emango digu:

> **(6-mean (Araba)) (standarddeviation (Araba));**

$$\frac{2}{7}$$ > **evalf (%);**

.2857142857

> **(5-mean (Bizkaia)) / standarddeviation (Bizkaia);**

.2974486033

> **(7-mean (Gipuzkoa)) / (standarddeviation (Gipuzkoa));**

$$\frac{13}{8781}\sqrt{8781}$$

> **evalf (%);** .1387303122

Beraz, Bizkaian 5 atera duena Araban 6 atera duenaren aurretik sailkatuko da, eta Araban 6 atera duena Gipuzkoan 7 atera duenaren aurretik.

BIBLIOGRAFIA

Xabier Alberdi Larizgoitia, Ibon Sarasola: *Euskal Estilo Libururantz*. EHUko Argitalpen Zerbitzua.

Martxel Ensunza, Jose Ramon Etxebarria, Jazinto Iturbe: *Zientzia eta teknikarako euskara*. UEU.

C.T.J. Dodson, E.A. González: *Experiments in Mathematics using Maple*. *Springer.*

E. Roanes Macías, E. Roanes Lozano: *Cálculos Matemáticos por ordenador con Maple*. V.5. Rubiños-1860, S.A.

César Pérez: *Cálculo simbólico y numérico con Mathematica*. Ra-ma.

N. Piskunov: *Kalkulu diferentziala eta integrala I eta II*. UEU.

José Colera, María José Oliveira, Rosario García: *Matematika II*. Anaya-Haritza.

José Colera, María José Oliveira, Rosario García: *Matematika Gizarte Zientziei Aplikatua II*. Anaya-Haritza.

José Colera, María José Oliveira, Rosario García, Santiago Fernández: *Matematika I*. Anaya-Haritza.

José Colera, María José Oliveira, Rosario García, Santiago Fernández: *Matematika Gizarte Zientziei Aplikatua II*. Anaya-Haritza.

Ángel de la Llave, Juan Carlos Peral: *Matemáticas 2*. Bruño.



Autor: Vicente Meavilla